Supplementary Material: SPSS MIXED Syntax for Evaluating Treatment and Covariate Effects with Partially Nested Data

This document provides syntax to implement models presented in Bauer, Sterba, and Hallfors (under review) for evaluating group-based interventions when control participants are ungrouped (and assumed to be independent). Specifically, SPSS MIXED syntax along with annotated excerpts of accompanying SPSS output, is provided for each of the three models fit by Bauer et al. in their empirical demonstration evaluating possible iatrogenic effects of the Reconnecting Youth (RY) preventive intervention program.

Data Description.

In the RY effectiveness trial, students were individually assigned to one of three study arms. In the Treatment Arm (RY), high-risk participants received RY treatment *administered in groups* (i.e. RY classes) composed by the experimenter. In the Control Arm (Control), high-risk participants were left *ungrouped*. In the Typical Arm (Typical), low-risk participants were also left *ungrouped*, serving as additional controls. The dependent variable in these analyses was deviant peer bonding (DPB) measured post-treatment. Participants were obtained from 9 high schools, and school was treated as a fixed factor in each of the fitted models (discussion on this point is provided in the original manuscript). As shown in the sixth column of Table A in the next section, Control, Typical, and Treatment subjects could come from the same or different schools. Other covariates common to each study arm were pre-treatment DPB, gender, age, and ethnicity. Regarding ethnicity, the sample was 6.5% Caucasian (N=107), 48.0% Hispanic (N=788), 11.9% African American (N=195), 26.5% Asian American (N=434), and 7.1% American Indian (N=117). Covariates unique to the RY condition were absences from the RY class, percent of students within the class that were female, and the average age of the class members.

Data Preparation.

Two data preparation steps are required specifically for these analyses. First, a group ID variable is required for all participants. A unique value must be assigned to each *group* in the treatment condition (i.e., each RY class) and to each *participant* in the ungrouped conditions. Table A shows an (artificial) section of the dataset with these group IDs assigned in column 1. Here is example SPSS code for assigning group IDs, where we assume the user has a pre-existing identification variable called *treatmentgroup*, which uniquely labels each group in which treatment was delivered, and another called *studentID*, which uniquely identifies the individual participants (without using the same numbers or labels as the *treatmentgroup* variable), and a preexisting character variable called *studyarm*, which specifies whether a given participant is from the Treatment, Control, or Typical arm of the study. (Note that *$sysmis* is a SPSS-defined code for missing data which we use simply to initiate a new variable).

```
compute GroupID=$sysmis.
 if studyarm='Treatment' GroupID=treatmentgroup.
 if studyarm='Control' GroupID=studentID.
 if studyarm='Typical' GroupID=studentID.
Execute.
```

Second, the values of any covariates relevant only for the treated (grouped) study arm (e.g., absences, percent female, and average age of class members) need to be set to an arbitrary non-missing value (e.g. -999) for individuals in the non-grouped study arm(s). This is shown in column 5 of Table A. Here is example SPSS code for assigning non-grouped individuals arbitrary non-missing values for an original group-level covariate called *percentfem*.

```
compute perfem=percentfem.
 if studyarm='Control' perfem=-999.
 if studyarm='Typical' perfem=-999.
Execute.
```

Table A.

GroupID	StudentID	Study Arm	Individual Covariate (e.g. *DPBpre*)	Group Covariate (e.g. *perfem*)	School
G1	1	Treatment	3.00	10.86	2
G1	2	Treatment	2.50	10.86	2
G1	3	Treatment	2.87	10.86	2
G1	4	Treatment	1.50	10.86	2
5	5	Control	0.50	-999	9
6	6	Control	0.00	-999	5
7	7	Typical	0.75	-999	9
8	8	Control	0.63	-999	2
9	9	Typical	1.99	-999	1
G2	10	Treatment	1.13	5.50	5
G2	11	Treatment	1.38	5.50	5
G2	12	Treatment	2.50	5.50	5
G2	13	Treatment	0.63	5.50	5
G2	14	Treatment	1.13	5.50	5
G3	15	Treatment	0.50	7.61	2
G3	16	Treatment	1.27	7.61	2
G3	17	Treatment	2.00	7.61	2

Model Fitting.

Model 1a: Evaluation of Treatment Effect with Assumption of Homoscedasticity for Individual Residuals

A key goal of Model 1a in Bauer et al.'s empirical demonstration was to test whether the average level of deviant peer bonding (measured post-treatment; *DPBpost*) differed between RY, Control and Typical participants, after including school as a fixed-factor control variable. *RY* is a dichotomous predictor indicating whether a student is assigned to RY-treatment (1), versus Control or Typical (0). *Typical* is a dichotomous predictor indicating whether a student is in assigned to Typical (1), versus Control or RY-Treatment (0). *School* is a nominal predictor with 9 categories, one for each school in the design. School is recoded into 8 dummy variables where

$School(c)_{ij}$ indicates whether a student i is in school c (coded 1) or not (coded 0). The level 1 equation for Model 1a (Equation 51 in Bauer et al.) is:

$$DPB_{ij} = \beta_{0j} + \beta_{1j} RY_{ij} + \beta_{2j} Typical_{ij} + \sum_{c=1}^{8} \beta_{(2+c)j} School(c)_{ij} + r_{ij}$$

In Model 1a, homoscedasticity for the residuals across arms of the study was assumed (i.e., $r_{ij} \sim N(0, \sigma^2)$ for all three conditions), however, this assumption will be relaxed and tested in Model 1b.

The level 2 equations for Model 1a (Equation 52 in Bauer et al.) are:

$$\begin{aligned}
\beta_{0j} &= \gamma_{00} \\
\beta_{1j} &= \gamma_{10} + u_{1j} \\
\beta_{2j} &= \gamma_{20} \\
\beta_{(2+c)j} &= \gamma_{(2+c)0}
\end{aligned}$$

Substituting the level 2 equations into the level 1 equation, the combined model equation for Model 1a (Equation 53 in Bauer et al.) is:

$$DPB_{ij} = \gamma_{00} + \gamma_{10} RY_{ij} + \gamma_{20} Typical_{ij} + \sum_{c=1}^{8} \gamma_{c0} School(c)_{ij} + u_{1j} RY_{ij} + r_{ij}$$

The syntax needed to fit this model using the MIXED procedure in SPSS is shown below, followed by a brief description of the primary statements.

```
MIXED
  DPBpost  BY school  WITH RY Typical
  /FIXED = school RY Typical  | SSTYPE(3)
  /RANDOM RY  | SUBJECT(GroupID) COVTYPE(VC)
  /CRITERIA = CIN(95) MXITER(100) MXSTEP(5) SCORING(1)
   SINGULAR(0.000000000001) HCONVERGE(0, ABSOLUTE) LCONVERGE(0, ABSOLUTE)
   PCONVERGE(0.000001, ABSOLUTE)
  /METHOD = REML
  /PRINT = SOLUTION TESTCOV LMATRIX
  /TEST = 'Cond' RY 1; Typical 1
  /TEST = 'Site' School .25 -.2 -.2 -.2 .25 -.2 -.2 .25 .25.
EXECUTE.
```

The first line of code contains the **MIXED** statement that calls SPSS's MIXED procedure. The second line contains the 'Variable List', where the outcome variable *DPBpost* must be listed first. Factors (i.e., nominal predictors, like *School*) are listed after the **BY** keyword. Covariates (which can include binary variables, here *RY* and *Typical*) are listed last after the **WITH** keyword. SPSS will automatically create 8 dummy variables to capture the 9 levels of the predictor *School*, with the last level of school (school 9) set to the reference category.

Next we need to specify the fixed and random effects in the model. Our model includes fixed effects of *RY*, *Typical*, and *School* on *DPBpost*, so these variables are included on the **/FIXED** statement (a fixed intercept is included by default). The **SSTYPE(n)** option on the **/FIXED** statement specifies the method for partitioning the sums of squares; n=3 is the default. This produces a table called "Type III Tests of Fixed Effects" in the output window. We additionally request that *t*-tests and standard errors for each fixed effect be output to a table called "Estimates of Fixed effects" by specifying **SOLUTION** on the **/PRINT** statement.

The **/RANDOM** statement is used to specify the random effects of the model. Thus, on this statement, we list predictors with random effects, i.e., effects that vary randomly across level-2 sampling units. For Model 1a, the only variable with a random effect is *RY*, so this is indicated here. By putting *RY* on the random statement we allow the effect of RY treatment on *DPBpost* to vary over RY classes and we account for within-class dependence in DPBpost scores. The level-2 sampling units are identified to SPSS using the **SUBJECT** option. Here, **SUBJECT(*GroupID*)**. Note that a random intercept is *not* included by default. **COVTYPE(VC)** requests the default (variance component) structure for random effects, which assumes all random effects are independent. The **TESTCOV** option on the **/PRINT** statement requests asymptotic standard errors, Wald-Z tests and confidence intervals for covariance parameters (producing a table of output called “Estimates of Covariance Parameters”). Note that SPSS does 2-tailed tests of variance parameters. (Bauer et al. reported 1-tailed p-values for these parameters, so the p-values generated by SPSS are two times larger).

The **/CRITERIA** statement requests particular details about the estimation algorithm. All of the options we requested on this statement are SPSS defaults, but we included them anyway so that the user could easily identify how to change them, if desired. *Therefore if the user does not want to deviate from the defaults, the user can omit this statement entirely.* We requested the default 95% confidence interval **CIN(95)**, and the default maximum number of iterations of 100, **MXITER(100)**, and the default maximum step-halving per iteration of 5, **MXSTEP(5)**, and the default of Fisher scoring up to iteration 1, **SCORING(1)**, and the default 10^{-12} as the value used to check singularity **SINGULAR(0.000000000001)**, and the default convergence criterion for parameter estimates of absolute change less than 10^{-6} , **PCONVERGE(0.000001, ABSOLUTE)**. We also requested that a log-likelihood convergence criteria not be used **LCONVERGE(0, ABSOLUTE)** and a Hessian convergence criteria not be used **HCONVERGE(0, ABSOLUTE)**; both are defaults.

Importantly, note that SPSS (Version 14) does not allow the user to specify how degrees of freedom should be calculated for testing fixed effects. Nor does SPSS (Version 14) adjust standard errors of fixed effects for the two sources of bias discussed in Bauer et al. under the section entitled “Testing Treatment Effects and Other Fixed Effects.” Rather, SPSS MIXED’s mandatory method of calculating degrees of freedom is the Fai and Cornelius (1996) Satterthwaite approximation (specified in SAS as ddfm= satterthwaite). This method was shown in computer simulations (see Schaalje, McBride & Fellingham, 2002) to have inflated Type I error compared to the Kenward-Roger method (available in SAS and advocated in Bauer et al.)—particularly under complex covariance structures, small samples, and unbalanced designs. In fact, as soon as the covariance structure departed from compound symmetric Schaalje et al. (2002) found that the Fai-Cornelius method produced approximate degrees of freedom that were too large (average .12 Type I error rate across four covariance structures). Also, when the sample size is small, the model-based standard errors are negatively biased (see Bauer et al. for explanation). The covariance structure of the present application does not greatly differ from compound symmetric and sample size is relatively large, so here the SPSS Fai-Cornelius method should produce similar results to SAS's Kenward-Rogers method. However, users should be aware that under less optimal circumstances, use of SPSS's Fai-Cornelius degrees of freedom calculation with unadjusted standard errors could bias inferences about fixed effects. Even in our example, readers will note some differences in the standard errors of the fixed effects between the SPSS output and the results reported by Bauer et al. in the manuscript and the accompanying SAS syntax appendix.

The /**METHOD** statement specifies the estimation method and the **REML** option calls the restricted maximum likelihood estimator for the model, which is the default. REML is selected because it typically provides less biased estimates of the variance components of the model than full information maximum likelihood, particularly when there are a small number of groups and/or large number of covariates.

The **/TEST** statement tests composite hypotheses involving linear combinations of fixed and/or random effects. Here, the first **/TEST** statement, labeled *Cond*, is used to test whether there is a significant DBP mean difference between participants in the three study arms. The second **/TEST** contrast statement, labeled *Site*, is used to test whether there is a significant DPB mean difference between the 4 schools at site A and the 5 schools at site B. SPSS outputs the *t*-test of single linear combinations to a table in the output window called “Contrast Estimates”. If multiple linear combinations are specified on the same **/TEST** statement (as in our *Cond* test), a joint *F*-test is also output to a table in the output window called “Tests of Contrasts.” By specifying the **LMATRIX** option on the **/PRINT** statement, we request that the matrix of coefficients for the contrasts be output so we can verify that we have indeed estimated the intended contrasts. These matrices are found in the “Contrast Coefficients” table of the output window.

Here, we provide a subset of the output produced by SPSS for Model 1a. Portions of output that can be (roughly) matched to values in the first column of Table 1 of Bauer, Sterba, and Hallfors (under review), and to interpretations on page 29-30 of this article, are indicated in bold font. In some cases the match is approximate, given that the results reported in the article were produced by SAS PROC MIXED using the Kenward-Rogers method of testing fixed effects. First we show the “Estimates of Fixed Effects” table from the output window:

Estimates of Fixed Effects(b)

Parameter	Estimate	Std. Error	df	t	Sig.	95% Confidence Interval	
						Lower Bound	Upper Bound
Intercept	1.111288	.067985	1166.576	16.346	.000	.977902	1.244675
[school=1]	.479431	.098531	1103.418	4.866	.000	.286101	.672761
[school=2]	.738674	.092314	751.589	8.002	.000	.557449	.919899
[school=3]	.714995	.124500	767.453	5.743	.000	.470594	.959396
[school=4]	.748538	.092456	932.588	8.096	.000	.567091	.929985
[school=5]	.221439	.096849	921.115	2.286	.022	.031369	.411509
[school=6]	.719005	.089873	867.978	8.000	.000	.542611	.895399
[school=7]	.824918	.123629	1284.386	6.673	.000	.582381	1.067456
[school=8]	.437642	.090055	744.114	4.860	.000	.260849	.614434
[school=9]	0(a)	0	.	.	.	.	.
RY	**.187951**	.071478	**68.254**	**2.629**	.011	.045328	.330574
Typical	**-.368997**	.052811	**1432.147**	**-6.987**	.000	-.472592	-.265403

a This parameter is set to zero because it is redundant. b Dependent Variable: DPBpost.

Each fixed effect is labeled by the variable to which it refers, and is accompanied by a *t*-test, confidence interval, degrees of freedom, and *p*-value. The fixed effect of *RY* represents the average difference in post-treatment DPB scores for individuals in the RY condition relative to the Control condition, whereas the fixed effect of *Typical* indicates the average difference in DPB scores for individuals in the Typical condition relative to the Control condition. Each school effect represents the mean difference in DPB scores between students of a particular school (1 through 8) relative to school 9, the reference school. A joint test of these eight fixed effects, representing the main effect of school, is provided by SPSS in the "Type 3 Tests of Fixed Effects" table of output:

Type III Tests of Fixed Effects(a)

Source	Numerator df	Denominator df	F	Sig.
Intercept	1	1451.274	1883.953	.000
school	**8**	**884.121**	**16.176**	.000
RY	1	68.254	6.914	.011
Typical	1	1432.147	48.821	.000

a Dependent Variable: DPBpost.

Note that the single degree of freedom tests reported here are redundant with the *t*-tests in the previous table of output.

Additionally, the "Tests of Contrasts" table in the output window provides an *F*-test of our '*Cond'* linear contrast (a multiple linear combination), identified with the label we placed in single quotes in the **/TEST** statement. The *F*-test of the '*Cond*' contrast (below) indicates that there is an overall mean difference in DPB between individuals in the RY, Control and Typical conditions. This represents a joint test of the *RY* and *Typical* fixed effects in the "Estimates of Fixed Effects" table presented previously.

Custom Hypothesis Test 1 (Cond)

Test of Contrasts(a)

Source	Numerator df	Denominator df	F	Sig.
Cond	2	183.146	**39.725**	.000

a Dependent Variable: DPBpost.

Another “Contrast Estimates” table in the output window (below) provides a *t*-test of the single linear combination in our *‘Site’* linear contrast; squaring the t-statistic yields an *F*-statistic which is comparable to that in Table 1 in Bauer et al. This *t*-test of the ‘*Site*’ contrast indicates that there is a significant overall mean difference in DPB for individuals from schools located at Site A versus individuals from schools located at Site B.

Custom Hypothesis Test 2 (Site)

Contrast Estimates(a,b)

Contrast	Estimate	Std. Error	df	Test Value	t	Sig.	95% Confidence Interval	
							Lower Bound	Upper Bound
L1	-.464598	.051348	**935.925**	0	**-9.048**	.000	-.565368	-.363828

a Site
b Dependent Variable: DPBpost.

The estimated variance components of the model are shown in the “Estimates of Covariance Parameters” table below:

Estimates of Covariance Parameters(a)

Parameter		Estimate	Std. Error	Wald Z	Sig.	95% Confidence Interval	
						Lower Bound	Upper Bound
Residual		**.788771**	.029513	**26.726**	.000	.732996	.848789
RY [subject = groupID]	Variance	**.052635**	.034744	**1.515**	.130	.014435	.191933

a Dependent Variable: DPBpost.

Note that the p-values here are produced by conducting two-tailed tests, whereas those reported in Bauer et al. were based on the 1-tailed tests conducted for variance parameters by SAS PROC MIXED. Dividing the p-values in the table above by 2 would produce equivalent 1-tailed p-values.

A final piece of output that will become relevant shortly is the -2*log-likelihood (or model deviance), which is produced by SPSS in the "Information Criteria" table:

Information Criteria(a)

-2 Restricted Log Likelihood	**3887.998**
Akaike's Information Criterion (AIC)	3891.998
Hurvich and Tsai's Criterion (AICC)	3892.006
Bozdogan's Criterion (CAIC)	3904.580
Schwarz's Bayesian Criterion (BIC)	3902.580

The information criteria are displayed in smaller-is-better forms.
a Dependent Variable: DPBpost.

We require the -2*log-likelihood, here reported as 3887.998, in order to evaluate the assumption of homoscedasticity of the individual residuals with our next model.

Model 1b: Allowing for Heteroscedasticity of the Individual Residuals

In Model 1b we allow the variance of the individual residuals to differ across the grouped and ungrouped conditions (i.e. RY condition versus Control and Typical conditions) to evaluate whether grouping participants results in the homogenization of behavior. To accomplish this, we place *RY* on the /**REPEATED** statement.

```
MIXED
  DPBpost  BY school  WITH RY Typical
  /FIXED = school RY Typical  | SSTYPE(3)
  /RANDOM RY  | SUBJECT(GroupID) COVTYPE(VC)
  /REPEATED  = RY |  SUBJECT(GroupID*StudentID) COVTYPE(DIAG)
  /CRITERIA = CIN(95) MXITER(100) MXSTEP(5) SCORING(1)
   SINGULAR(0.000000000001) HCONVERGE(0, ABSOLUTE) LCONVERGE(0, ABSOLUTE)
   PCONVERGE(0.000001, ABSOLUTE)
  /METHOD = REML
  /PRINT = SOLUTION TESTCOV LMATRIX
  /TEST = 'Cond' RY 1; Typical 1
  /TEST = 'Site' School .25 -.2 -.2 -.2 .25 -.2 -.2 .25 .25.
EXECUTE.
```

In general, the **/REPEATED** statement provides access to the variances and covariances of the individual (level 1) residuals. In Model 1a, these residuals were assumed to be independent and of constant variance across the grouped (Treatment) and ungrouped (Control and Typical) arms of the study. With Model 1b, we relax the assumption of constant variance, requesting that a different variance be estimated for the residuals of individuals in the RY condition relative to the two ungrouped conditions. We add the **SUBJECT** option to the **/REPEATED** statement to specify the heteroscedastic residuals, indicating that the residual variance will differ by students (within groups) across levels of the *RY* variable. The **COVTYPE(type)** option is set to the default **COVTYPE(DIAG).**

The primary output of interest for this model is the "Covariance Parameter Estimates" table, which provides our new residual variance estimates:

Estimates of Covariance Parameters(a)

Parameter		Estimate	Std. Error	Wald Z	Sig.	95% Confidence Interval	
						Lower Bound	Upper Bound
Repeated Measures	Var: [RY=0]	**.782399**	.032865	23.807	.000	.720566	.849538
	Var: [RY=1]	**.813262**	.066985	12.141	.000	.692023	.955741
RY [subject = groupID]	Variance	.049807	.035368	1.408	.159	.012383	.200327

a Dependent Variable: DPBpost.

As can be seen, the residual variance for the *RY* condition (labeled ‘RY=1’) is higher than the residual variance for the other two conditions (labeled ‘RY=0’), in contradiction to the hypothesis of within-group homogenization. We are also interested in the -2*log-likelihood of this model which, when contrasted with the -2*log-likelihood of Model 1a, will provide a likelihood ratio test of the assumption of homoscedasticity for the individual residuals. We obtain the -2*log likelihood of this model the “Information Criteria” table, shown here:

Information Criteria(a)

-2 Restricted Log Likelihood	**3887.822**
Akaike's Information Criterion (AIC)	3893.822
Hurvich and Tsai's Criterion (AICC)	3893.838
Bozdogan's Criterion (CAIC)	3912.694
Schwarz's Bayesian Criterion (BIC)	3909.694

The information criteria are displayed in smaller-is-better forms.
a Dependent Variable: DPBpost.

The difference in fit between the two models follows a chi-square distribution with degrees of freedom equal to the difference in the number of parameters (in this case, one). The likelihood ratio test can be computed directly in SPSS using the following syntax:

```
DATA LIST FREE/ dev1 dev2.
BEGIN DATA
  3888.0 3887.8
END DATA.

COMPUTE chi = dev1-dev2.
COMPUTE df = 1.
COMPUTE pvalue = 1- cdf.chisq(chi,df) .
EXECUTE .
LIST chi df pvalue.

DATASET ACTIVATE RYdat WINDOW=FRONT.
EXECUTE .
```

After the BEGIN DATA command, the -2*Log-likelihoods of Models 1A and 1B are entered. The COMPUTE commands then calculate the likelihood ratio test. Note that the **`CDF.CHISQ`** function returns the probability of obtaining a likelihood ratio less than or equal to the one we obtained. Subtracting from 1 then yields the probability of finding a likelihood ratio this large or larger, under the null hypothesis of no difference. The LIST command prints the results to the output window. Finally, the DATASET ACTIVATE command resets the active dataset to RYdat for the models that will be run next. The result is

List

chi	df	pvalue
.20	1.00	.65

We thus find that the individual residual variances are not sufficiently different between the study arms to warrant rejection of the homoscedastic model (Model 1a).

Model 2: Adding Common Covariates

Model 2 includes four additional level-1 fixed effect covariates measured in all three study arms: baseline delinquency (*DPBpre*), *ethnicity*, *sex*, and age (*ageyrs*). We will call these “common” covariates to emphasize that they are measured in all study arms. Model 2 assumes that the relationship between each common covariate and DPBpost is constant over treatment groups. Since these covariates enter the model only through the addition of fixed effects, we forgo writing out the model equations here, as this becomes somewhat tedious. The additional fixed effects of the covariates are produced by first adding these variables to the 'Variable List' on the line following the **MIXED** statement, and, second, adding these variables to the **/FIXED** statement, as shown below. We also need to declare blanks as missing values for ethnicity in our particular dataset, using the **MISSING VALUES** syntax. The rest of the syntax is identical to Model 1a.

```
MISSING VALUES ethnicity (" ").
EXECUTE .

MIXED
  DPBpost  BY school ethnicity sex WITH RY Typical DPBpre ageyrs
  /FIXED = school ethnicity sex RY Typical DPBpre ageyrs | SSTYPE(3)
  /RANDOM RY  | SUBJECT(GroupID) COVTYPE(VC)
  /CRITERIA = CIN(95) MXITER(100) MXSTEP(5) SCORING(1)
   SINGULAR(0.000000000001) HCONVERGE(0, ABSOLUTE) LCONVERGE(0, ABSOLUTE)
   PCONVERGE(0.000001, ABSOLUTE)
  /METHOD = REML
  /PRINT = SOLUTION TESTCOV LMATRIX
  /TEST = 'Cond' RY 1; Typical 1
  /TEST = 'Site' School .25 -.2 -.2 -.2 .25 -.2 -.2 .25 .25.
EXECUTE.
```

Selections from the output are shown below. Bold portions of output from Model 2 can be matched to values in the second column of Table 1 of Bauer et al. and to interpretations on page 30 of Bauer et al. We first consider the estimates of fixed effects:

Estimates of Fixed Effects(b)

Parameter	Estimate	Std. Error	df	t	Sig.	95% Confidence Interval	
						Lower Bound	Upper Bound
Intercept	1.229539	.388225	1451.359	3.167	.002	.467997	1.991080
[school=1]	.236659	.089876	1113.456	2.633	.009	.060314	.413004
[school=2]	.307722	.100187	930.668	3.071	.002	.111102	.504341
[school=3]	.242445	.125810	831.223	1.927	.054	-.004498	.489388
[school=4]	.385466	.101930	1096.465	3.782	.000	.185465	.585467
[school=5]	.049954	.085499	863.690	.584	.559	-.117855	.217764
[school=6]	.310904	.096825	1108.324	3.211	.001	.120923	.500884
[school=7]	.477085	.128155	1371.982	3.723	.000	.225683	.728486
[school=8]	.170158	.078358	581.865	2.172	.030	.016258	.324057
[school=9]	0(a)	0	.	.	.	.	.
[ethnicity=Ameri]	.021608	.113034	1454.520	.191	.848	-.200119	.243335
[ethnicity=Asian]	-.234572	.089869	1453.286	-2.610	.009	-.410858	-.058286
[ethnicity=Black]	-.017189	.110986	1454.518	-.155	.877	-.234898	.200520
[ethnicity=Latin]	-.156945	.097619	1453.365	-1.608	.108	-.348434	.034544
[ethnicity=White]	0(a)	0	.	.	.	.	.
[sex=1]	**.064518**	.040926	**1454.993**	**1.576**	.115	-.015762	.144798
[sex=2]	0(a)	0	.	.	.	.	.
RY	**.138161**	.056811	**79.399**	**2.432**	.017	.025090	.251232
Typical	**-.106282**	.048462	**1427.832**	**-2.193**	.028	-.201347	-.011216
DPBpre	**.495840**	.023218	**1444.439**	**21.356**	.000	.450297	.541384
ageyrs	**-.034982**	.024947	**1450.281**	**-1.402**	.161	-.083918	.013955

a This parameter is set to zero because it is redundant.
b Dependent Variable: DPBpost.

For the categorical independent variables, multi-degree of freedom tests of main effects are also available in the "Type 3 Tests of Fixed Effects" table.

Type III Tests of Fixed Effects(a)

Source	Numerator df	Denominator df	F	Sig.
Intercept	1	1450.762	13.869	.000
school	8	**753.208**	**3.069**	.002
ethnicity	4	**1453.402**	**4.078**	.003
sex	1	1454.993	2.485	.115
RY	1	79.399	5.914	.017
Typical	1	1427.832	4.810	.028
DPBpre	1	1444.439	456.090	.000
ageyrs	1	1450.281	1.966	.161

a Dependent Variable: DPBpost.

Overall, only the *t*-test of the fixed effect of *DBPpre* in the "Estimates of Fixed Effects" table and the overall *F*-test of the fixed effect of *ethnicity* and *school* in the "Type 3 Tests of Fixed Effects" table uniquely explain additional variability in DBPpost. The differences between the three study arms, however, are maintained after controlling for the covariates. Our two linear contrasts also remain significant as shown below.

Custom Hypothesis Test 1 (Cond)

Test of Contrasts(a)

Source	Numerator df	Denominator df	F	Sig.
Cond	2	220.556	8.545	**.000**

a Dependent Variable: DPBpost.

Custom Hypothesis Test 2 (Site)

Contrast Estimates(a,b)

Contrast	Estimate	Std. Error	df	Test Value	t	Sig.	95% Confidence Interval	
							Lower Bound	Upper Bound
L1	-.230531	.064910	1133.133	0	-3.552	**.000**	-.357889	-.103174

a Site
b Dependent Variable: DPBpost.

The other results of interest from this model are the variance component estimates, shown here:

Estimates of Covariance Parameters(a)

Parameter		Estimate	Std. Error	Wald Z	Sig.	95% Confidence Interval	
						Lower Bound	Upper Bound
Residual		**.592458**	.022247	26.631	.000	.550421	.637706
RY [subject = groupID]	Variance	**.014204**	.019951	.712	.477	.000905	.222859

a Dependent Variable: DPBpost.

As can be seen by comparison to the variance components obtained from Model 1a, the inclusion of the four pre-existing difference covariates explains almost all of the across-group variability in the effect of RY treatment on DPB. This is evidenced by the nonsignificant, near-zero estimate for the variance of the random slope of RY-treatment on DPB (labeled *RY* in the “Estimates of Covariance Parameters” output table.) This represents a decrease in the variance of the random slope from .05263 in Model 1a to .01420 here in Model 2.

<u>Model 3: Adding Covariates Unique to Grouped Condition</u>

The syntax for Model 3 adds two group-level predictors to explain why treatment may have been more detrimental for some RY-treatment groups than others. They are the mean age of the group (*meanage*), and the percentage of females in the group (*perfem*). In addition, the individual level predictor, number of absences from the RY class, was included to assess possible dosage effects. These predictors are only relevant for members of RY classes and are, in essence, undefined for participants in the other study arms. For this reason, they should only impact the DPB scores of students in the RY study arm and should have no effect for the other two study arms. To meet these constraints, these predictors are entered through interactions with the RY variable (see extended discussion in Bauer, Sterba & Hallfors, under review). No main effects are included because the effect is necessarily null in the control conditions. Additionally, it is worth repeating that these covariates are coded -999 for all participants in the control or typical conditions (an arbitrary value that is *not* declared as a missing data code) so that these participants will not be listwise deleted from the analysis (as discussed in the Data Preparation section above). The syntax for fitting the model is:

```
MIXED
  DPBpost  BY school ethnicity sex WITH RY Typical DPBpre ageyrs meanage absences perfem
  /FIXED = school ethnicity sex RY Typical DPBpre ageyrs RY*meanage RY*absences RY*perfem | SSTYPE(3)
  /RANDOM RY  | SUBJECT(GroupID) COVTYPE(VC)
  /CRITERIA = CIN(95) MXITER(100) MXSTEP(5) SCORING(1)
   SINGULAR(0.000000000001) HCONVERGE(0, ABSOLUTE) LCONVERGE(0, ABSOLUTE)
   PCONVERGE(0.000001, ABSOLUTE)
  /METHOD = REML
  /PRINT = SOLUTION TESTCOV LMATRIX
  /TEST = 'Cond' RY 1; Typical 1
  /TEST = 'Site' School .25 -.2 -.2 -.2 .25 -.2 -.2 .25 .25.
EXECUTE.
```

The syntax for Model 3 is identical to the syntax from Model 2 except for the three new fixed effects, that are (1) listed on the 'Variable List' line and (2) are listed on the **/FIXED** statement as *RY*meanage*, *RY*perfem*, and *RY*absences*. These new fixed effects are cross-products that can be declared explicitly within the given syntax in MIXED rather than via COMPUTE commands. Notice that we did not enter the main effects of these predictors on either the 'Variable List' or the **/FIXED** statements**.**

The estimates for the fixed effects are shown here, with bolded entries appearing in column 3 of Table 1 in Bauer et al.:

Estimates of Fixed Effects(b)

Parameter	Estimate	Std. Error	df	t	Sig.	95% Confidence Interval	
						Lower Bound	Upper Bound
Intercept	1.130725	.393880	1423.274	2.871	.004	.358078	1.903372
[school=1]	.239961	.089897	1049.452	2.669	.008	.063563	.416359
[school=2]	.302168	.102086	1073.597	2.960	.003	.101858	.502478
[school=3]	.246379	.125973	764.382	1.956	.051	-.000916	.493673
[school=4]	.384064	.102188	1030.164	3.758	.000	.183543	.584585
[school=5]	.045087	.085529	800.800	.527	.598	-.122801	.212975
[school=6]	.314812	.096824	1025.275	3.251	.001	.124816	.504808
[school=7]	.479727	.128114	1322.438	3.745	.000	.228398	.731055
[school=8]	.164116	.078649	532.143	2.087	.037	.009617	.318616
[school=9]	0(a)	0	.	.	.	.	.
[ethnicity=Ameri]	.001994	.113667	1429.113	.018	.986	-.220979	.224966
[ethnicity=Asian]	-.244776	.090095	1429.032	-2.717	.007	-.421509	-.068043
[ethnicity=Black]	-.033255	.111687	1429.047	-.298	.766	-.252344	.185833

[ethnicity=Latin]	-.162263	.098304	1428.760	-1.651	.099	-.355098	.030573
[ethnicity=White]	0(a)	0	.	.	.	.	.
[sex=1]	.056277	.041569	1409.160	1.354	.176	-.025268	.137821
[sex=2]	0(a)	0	.	.	.	.	.
RY	.139191	.057446	64.693	2.423	.018	.024453	.253928
Typical	-.106589	.048426	1404.601	-2.201	.028	-.201584	-.011593
DPBpre	.492202	.023416	1413.029	21.020	.000	.446267	.538137
ageyrs	-.027209	.025364	1418.866	-1.073	.284	-.076964	.022546
RY * meanage	**-.136239**	.133176	**33.376**	**-1.023**	**.314**	-.407072	.134594
RY * absences	**.001584**	.003052	**662.045**	**.519**	**.604**	-.004408	.007576
RY * PerFem	**-.001280**	.002649	**34.708**	**-.483**	**.632**	-.006660	.004099

a This parameter is set to zero because it is redundant.
b Dependent Variable: DPBpost.

Considering first the group-level predictors, we can see that having more girls (*RY*perfem)* and older students in the group (*RY*meanage*) slightly decreases DPB post scores, though neither effect is significant. Additionally, more absences (*RY*absences)* are associated with higher DPBpost scores. Other output (i.e., covariance parameters, contrasts, and multi-degree of freedom tests) is identical in form to Models 1a and 2 and so is not presented here.

References

Schaalje, G.B., McBride, J.J. and Fellingham, G.W. (2002). Adequacy of approximations to distributions of test statistics in complex mixed linear models. *Journal of Agricultural, Biological and Environmental Statistics, 7*, 512-524.